
サンプル
プログラム

xy_actuator_001.F2K
サンプル
点データ

xy_actuator_001.P2K

設定したXYのポイントを順番に移動する

1) スタートSWを押して動作開始

2) 初めに原点復帰

3) ポイント1から9まで順番に移動し、ポイント1に戻る

4) 3)の動作を3回繰り返し、ポイント1に停止

5) 途中で停止したい場合は、ストップSWを押す

　 再開する場合は、スタートSWを押す

ステッピングモータ ｽﾃｯﾌﾟ角1.8°、最大ﾊﾟﾙｽ周波数3kHz

マイクロステップドライバ 分解能 5分割

ﾓｰﾀ1回転あたりのﾊﾟﾙｽ数 360°/1.8°×5＝1000パルス

ベルトピッチ 2mm

プーリの歯数 32歯

1軸の有効移動量 約170mm

リミットSW X軸、Y軸の原点

サンプルプログラム例

例題 XY軸直交アクチュエータを動作させる

使用機器の仕様

動作概要 システム構成イメージ

動作イメージ 動作フローチャート

【点データの設定方法】

サンプル点データ「xy_actuator_001.P2K」をご参照ください。

1）点データの設定方法は下記になります。「VOID」を入力すると、引数は無効となります。

   点データは、4バイト整数となります。

点データ　　　座標データ

　　　№　X軸　Y軸　U軸　Z軸

SETP 4 100 1350 VOID VOID ←点データ4 X:100  Y:1350 U:無効 Z:無効

SETP 100 1500 2000 300　-500 ←点データ100 X:1500  Y:2000 U:300 Z:-500

2)サンプル点データ「xy_actuator_001.P2K」の記述は下記となります。

SETP 1 100 100 VOID VOID ←点データ1 X:100  Y:100  U:無効 Z:無効

SETP 2 1350 100 VOID VOID ←点データ2 X:1350 Y:100  U:無効 Z:無効

SETP 3 2600 100 VOID VOID ←点データ3 X:2600 Y:100  U:無効 Z:無効

SETP 4 100 1350 VOID VOID ←点データ4 X:100  Y:1350 U:無効 Z:無効

SETP 5 1350 1350 VOID VOID ←点データ5 X:1350 Y:1350 U:無効 Z:無効

SETP 6 2600 1350 VOID VOID ←点データ6 X:2600 Y:1350 U:無効 Z:無効

SETP 7 100 2600 VOID VOID ←点データ7 X:100  Y:2600 U:無効 Z:無効

SETP 8 1350 2600 VOID VOID ←点データ8 X:1350 Y:2600 U:無効 Z:無効

SETP 9 2600 2600 VOID VOID ←点データ9 X:2600 Y:2600 U:無効 Z:無効

xy_actuator_001.F2K
/******************************************************************************
/*例題                        ：XY軸直交アクチュエータを動作させる
/*サンプルプログラム          ：xy_actuator_001.F2K
/*サンプル点データ            ：xy_actuator_001.P2K
/*
/*ステッピングモータ仕様      ：ステップ角1.8°、最大パルス周波数3kHz
/*マイクロステップドライバ仕様：分解能 5分割
/*モータ1回転あたりのパルス数 ：360°/1.8°×5＝1000パルス
/*ベルトピッチ                ：2mm
/*プーリの歯数                ：32歯
/*1軸の有効移動量             ：約170mm
/*リミットSW                  ：X軸、Y軸の原点
/******************************************************************************

/***** 初期処理 *****

PG 17                           /*MPC-1200のPGアサイン

ACCEL ALL_A 3000 500 100        /*XY軸、最高速3kpps、加速度500パルス
                                /*最低速100pps
FEED ALL_A 100                  /*全軸、速度100%

STSW=0                          /*START SWフラグクリア
SPSW=0                          /*STOP SWフラグクリア

/***** タスク設定 *****
FORK 1 *MAIN                    /*タスク1:メイン処理
FORK 2 *SW                      /*タスク2:SW処理
END

/***** メイン処理 *****
*MAIN                           /*タスク1
PG 17                           /*MPC-1200のPGアサイン

WAIT STSW==1                    /*START SWフラグがON(=1)なら動作開始
GOSUB *XY_HOME                  /*XY軸原点復帰

DO

LP=3                            /*繰り返し回数：3回
PS=9                            /*ポイント数：9

 FOR c=1 TO LP                  /*繰り返し回数まで繰り返し
  WAIT STSW==1                  /*START SWフラグがON(=1)なら動作開始

  FOR i=1 TO PS                 /*ポイント1～9まで繰り返し
   WAIT STSW==1                 /*START SWフラグがON(=1)なら動作開始
   MOVS X(i) Y(i) VOID VOID     /*ポイントの座標まで移動
   WAIT RR(ALL_A)==0            /*パルスが0になるまで待つ
   TIME 100                     /*100msec待ち(100×1msec)

   WAIT SPSW==0                 /*STOP SWフラグがON(=1)なら動作停止
  NEXT i

  MOVS X(1) Y(1) VOID VOID      /*ポイント1に戻る
  WAIT RR(ALL_A)==0             /*パルスが0になるまで待つ
  TIME 100                      /*100msec待ち(100×1msec)
 NEXT c

 STSW=0                         /*START SWフラグクリア
 SPSW=0                         /*STOP SWフラグクリア
LOOP

/***** SW処理 *****
*SW                             /*タスク2
DO
 TIME 10                        /*10msec秒待ち(10×1msec)

ページ(1)



xy_actuator_001.F2K
 IF SW(192)==1 THEN             /*START SWがON(=1)なら動作開始
  STSW=1                        /*START SWフラグセット
  SPSW=0                        /*STOP SWフラグクリア
  WAIT SW(192)==0               /*START SWを離すまで待つ
 END_IF

 IF SW(193)==1 THEN             /*STOP SWがON(=1)なら動作停止
  STSW=0                        /*START SWフラグクリア
  SPSW=1                        /*STOP SWフラグセット
  WAIT SW(193)==0               /*STOP SWを離すまで待つ
 END_IF
 SWAP                           /*強制スワップ(実行タスク切替)
LOOP

/***** XY軸原点復帰 *****
*XY_HOME
IF SW(200)==1 THEN              /*X軸原点リミットSWがON(=1)なら一旦退避移動
 RMVS X_A 100                   /*CW方向に100パルス移動
END_IF
IF SW(201)==1 THEN              /*Y軸原点リミットSWがON(=1)なら一旦退避移動
 RMVS Y_A 100                   /*CW方向に100パルス移動
END_IF
WAIT RR(ALL_A)==0               /*XY軸、パルスが0になるまで待つ

FEED ALL_A 20                   /*全軸、速度20%に速度を落とす
SHOM &H05                       /*XY軸SD入力(bit2,0)を無効(=1)にする
HOME X_A|Y_A NEG_L              /*XY軸原点リミットSWがON(=1)までCCW方向に原点復帰移動
WAIT RR(ALL_A)==0               /*XY軸、パルスが0になるまで待つ
TIME 500                        /*500msec待ち(500×1msec)

CLRPOS                          /*全軸、位置カウンタクリア
FEED ALL_A 100                  /*全軸、速度100%に速度を戻す
TIME 1000                       /*1秒待ち(1000×1msec)
RETURN

xy_actuator_001.P2K
SETP 1 100 100 VOID VOID
SETP 2 1350 100 VOID VOID
SETP 3 2600 100 VOID VOID
SETP 4 100 1350 VOID VOID
SETP 5 1350 1350 VOID VOID
SETP 6 2600 1350 VOID VOID
SETP 7 100 2600 VOID VOID
SETP 8 1350 2600 VOID VOID
SETP 9 2600 2600 VOID VOID

